NAG Fortran Library Routine Document

G05LQF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05LQF generates a vector of pseudo-random numbers from an exponential mix distribution composed of m exponential distributions each having a mean a_i and weight w_i.

2 Specification

```
SUBROUTINE G05LQF(NMIX, A, WGT, N, X, IGEN, ISEED, IFAIL)
INTEGER           NMIX, N, IGEN, ISEED(4), IFAIL
real              A(NMIX), WGT(NMIX), X(*)
```

3 Description

The distribution has PDF (probability density function)

$$f(x) = \begin{cases} \displaystyle\sum_{i=1}^{m} \frac{1}{a_i} w_i e^{-x/a_i} & \text{if } x > 0, \\ 0 & \text{otherwise,} \end{cases}$$

where $\sum_{i=1}^{m} w_i = 1$ and $a_i > 0$, $w_i \geq 0$.

The routine returns the values x_i by selecting, with probability w_j, random variates from an exponential distribution with parameter a_j.

One of the initialisation routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05LQF.

4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison-Wesley

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin

5 Parameters

1: NMIX – INTEGER *Input*

On entry: the number, m, of exponential distributions in the mix.

Constraint: NMIX ≥ 1.

2: A(NMIX) – ***real*** array *Input*

On entry: the m parameters a_i for the m exponential distributions in the mix.

Constraint: A$(I) > 0.0$, for $I = 1, \ldots,$ NMIX.

3: WGT(NMIX) – ***real*** array *Input*

On entry: the m weights w_i for the m exponential distributions in the mix.

Constraints:

$\sum_{I=1}^{\text{NMIX}} \text{WGT}(I) = 1.0$

$\text{WGT}(I) \geq 0.0$, for $I = 1, \ldots, \text{NMIX}$

4: N – INTEGER *Input*

On entry: the number, n, of pseudo-random numbers to be generated.

Constraint: $\text{N} \geq 0$.

5: X(*) – ***real*** array *Output*

Note: the dimension of the array X must be at least $\max(1, \text{N})$.

On exit: the n pseudo-random numbers from the specified exponential mix distribution.

6: IGEN – INTEGER *Input*

On entry: must contain the identification number for the generator to be used to return a pseudo-random number and should remain unchanged following initialisation by a prior call to one of the routines G05KBF or G05KCF.

7: ISEED(4) – INTEGER array *Input/Output*

On entry: contains values which define the current state of the selected generator.

On exit: contains updated values defining the new state of the selected generator.

8: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: $\text{IFAIL} = 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry $\text{IFAIL} = 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$\text{IFAIL} = 1$

On entry, $\text{NMIX} \leq 0$.

$\text{IFAIL} = 2$

On entry, $\text{A}(I) \leq 0.0$ for at least one $\text{A}(I)$.

$\text{IFAIL} = 3$

On entry, $\text{WGT}(I) < 0.0$ for at least one $\text{WGT}(I)$.

$\text{IFAIL} = 4$

On entry, $\sum_{I=1}^{\text{NMIX}} \text{WGT}(I) \neq 1.0$.

$\text{IFAIL} = 5$

On entry, $\text{N} < 0$.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

The example program prints the first five pseudo-random real numbers from an exponential mix distribution comprising three exponential distributions with parameters $a_1 = 1.0$, $a_2 = 5.0$ and $a_3 = 2.0$, and with respective weights 0.5, 0.3 and 0.2. The numbers are generated by a single call to G05LQF, after initialisation by G05KBF.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G05LQF Example Program Text
*     Mark 20 Release. NAG Copyright 2001.
*     .. Parameters ..
      INTEGER          NOUT, M, NMIX
      PARAMETER        (NOUT=6,M=5,NMIX=3)
*     .. Local Scalars ..
      INTEGER          IFAIL, IGEN
*     .. Local Arrays ..
      real             A(NMIX), WGT(NMIX), X(M)
      INTEGER          ISEED(4)
*     .. External Subroutines ..
      EXTERNAL         G05KBF, G05LQF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05LQF Example Program Results'
      WRITE (NOUT,*)
*     Initialise the seed to a repeatable sequence
      ISEED(1) = 1762543
      ISEED(2) = 9324783
      ISEED(3) = 42344
      ISEED(4) = 742355
*     IGEN identifies the stream.
      IGEN = 1
      CALL G05KBF(IGEN,ISEED)
*
      A(1) = 1.0e0
      A(2) = 5.0e0
      A(3) = 2.0e0
*
      WGT(1) = 0.5e0
      WGT(2) = 0.3e0
      WGT(3) = 0.2e0

      IFAIL = 0
      CALL G05LQF(NMIX,A,WGT,M,X,IGEN,ISEED,IFAIL)
*
      WRITE (NOUT,99999) X
      STOP
*
99999 FORMAT (1X,F10.4)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05LQF Example Program Results

     0.8275
     1.0723
     0.9284
     5.4923
     0.1827
```